

Exploratory and Attempted Conformatory Factor Analysis for a Non-identified Two-factor Model

```
/* fac2.sas */
options linesize=79 pagesize=200noovp formdlim='_';
title 'Confirmatory and Exploratory factor analysis with simulated data';
title2 'Non-identified 2-factor model';

data latent1; /* Variables load only on one factor, but we don't know it. */
  n = 500; iseed = 51493;

  /* True factor loadings (All communalities = .25) */
  L11 = .5; L12 = 0;
  L21 = .5; L22 = 0;
  L31 = .5; L32 = 0;
  L41 = .5; L42 = 0;
  L51 = 0; L52 = .5;
  L61 = 0; L62 = .5;
  L71 = 0; L72 = .5;
  L81 = 0; L82 = .5;

  /* Variances of error terms will all equal 0.75*/
  v1 = 1 - L11**2 - L12**2;
  v2 = 1 - L21**2 - L22**2;
  v3 = 1 - L31**2 - L32**2;
  v4 = 1 - L41**2 - L42**2;
  v5 = 1 - L51**2 - L52**2;
  v6 = 1 - L61**2 - L62**2;
  v7 = 1 - L71**2 - L72**2;
  v8 = 1 - L81**2 - L82**2;

  do i=1 to n;
    /* Factors are independent standard normal */
    F1 = rannor(iseed); F2 = rannor(iseed);
    /* Observed variables all have variance one */
    X1 = L11*F1 + L12*F2 + sqrt(v1)*rannor(iseed);
    X2 = L21*F1 + L22*F2 + sqrt(v2)*rannor(iseed);
    X3 = L31*F1 + L32*F2 + sqrt(v3)*rannor(iseed);
    X4 = L41*F1 + L42*F2 + sqrt(v4)*rannor(iseed);
    X5 = L51*F1 + L52*F2 + sqrt(v5)*rannor(iseed);
    X6 = L61*F1 + L62*F2 + sqrt(v6)*rannor(iseed);
    X7 = L71*F1 + L72*F2 + sqrt(v7)*rannor(iseed);
    X8 = L81*F1 + L82*F2 + sqrt(v8)*rannor(iseed);
    output; /* Create a case */
  end;
  keep x1-x8;
```

```

proc calis pcorr; /* Print correlation matrix */
title3 'Try to fit non-identified 2-factor model with proc calis';
var X1-X8;          /* Manifest vars are in the data set */
lineqs             /* Simultaneous equations, separated by commas */
  X1 = lambda11 F1 + lambda12 F2 + delta1,
  X2 = lambda21 F1 + lambda22 F2 + delta2,
  X3 = lambda31 F1 + lambda32 F2 + delta3,
  X4 = lambda41 F1 + lambda42 F2 + delta4,
  X5 = lambda51 F1 + lambda52 F2 + delta5,
  X6 = lambda61 F1 + lambda62 F2 + delta6,
  X7 = lambda71 F1 + lambda72 F2 + delta7,
  X8 = lambda81 F1 + lambda82 F2 + delta8;

std               /* Variances (not standard deviations) */
  F1 = 1 , F2 = 1,
  delta1-delta8 = theta1-theta8;
bounds 0.0 < theta1-theta8; /* Variances are positive */

proc factor method=ML rotate=varimax;
title3 'Estimate Loadings with exploratory factor analysis';
var X1 - X8;

/* Try starting proc calis at the rotated solution */

proc calis;
title3 'Start at rotated solution: Objective Function = 0.046133184?';
var X1-X8;          /* Manifest vars are in the data set */
lineqs             /* Simultaneous equations, separated by commas */
  X1 = lambda11 (0.44623) F1 + lambda12 (-0.00138) F2 + delta1,
  X2 = lambda21 (0.61703) F1 + lambda22 (-0.10112) F2 + delta2,
  X3 = lambda31 (0.49899) F1 + lambda32 (-0.05083) F2 + delta3,
  X4 = lambda41 (0.47995) F1 + lambda42 (0.12903) F2 + delta4,
  X5 = lambda51 (0.03523) F1 + lambda52 (0.49182) F2 + delta5,
  X6 = lambda61 (-0.04097) F1 + lambda62 (0.47610) F2 + delta6,
  X7 = lambda71 (0.04314) F1 + lambda72 (0.43286) F2 + delta7,
  X8 = lambda81 (-0.07363) F1 + lambda82 (0.58452) F2 + delta8;

std               /* Variances (not standard deviations) */
  F1 = 1 , F2 = 1,
  delta1-delta8 = theta1-theta8;
bounds 0.0 < theta1-theta8; /* Variances are positive */

```

```

data latent2; /* Variables load on both factors */
  n = 500; iseed = 51493;

  /* True factor loadings (All communalities = .5^2 + .8^2 = 0.89) */
  L11 = .5; L12 = -.8;
  L21 = .5; L22 = -.8;
  L31 = .5; L32 = -.8;
  L41 = .5; L42 = -.8;
  L51 = .8; L52 = .5;
  L61 = .8; L62 = .5;
  L71 = .8; L72 = .5;
  L81 = .8; L82 = .5;

  /* Variances of error terms all equal 0.11 */
  v1 = 1 - L11**2 - L12**2;
  v2 = 1 - L21**2 - L22**2;
  v3 = 1 - L31**2 - L32**2;
  v4 = 1 - L41**2 - L42**2;
  v5 = 1 - L51**2 - L52**2;
  v6 = 1 - L61**2 - L62**2;
  v7 = 1 - L71**2 - L72**2;
  v8 = 1 - L81**2 - L82**2;

  do i=1 to n;
    /* Factors are independent standard normal */
    F1 = rannor(iseed); F2 = rannor(iseed);
    /* Observed variables all have variance one */
    X1 = L11*F1 + L12*F2 + sqrt(v1)*rannor(iseed);
    X2 = L21*F1 + L22*F2 + sqrt(v2)*rannor(iseed);
    X3 = L31*F1 + L32*F2 + sqrt(v3)*rannor(iseed);
    X4 = L41*F1 + L42*F2 + sqrt(v4)*rannor(iseed);
    X5 = L51*F1 + L52*F2 + sqrt(v5)*rannor(iseed);
    X6 = L61*F1 + L62*F2 + sqrt(v6)*rannor(iseed);
    X7 = L71*F1 + L72*F2 + sqrt(v7)*rannor(iseed);
    X8 = L81*F1 + L82*F2 + sqrt(v8)*rannor(iseed);
    output; /* Create a case */
  end;

proc factor method=ML rotate=varimax;
title3 'Expl. factor analysis when all vars really load on both factors';
var X1 - X8;

```

Confirmatory and Exploratory factor analysis with simulated data 1
 Non-identified 2-factor model
 Try to fit non-identified 2-factor model with proc calis
 22:17 Wednesday, April 5, 2006

The CALIS Procedure
 Covariance Structure Analysis: Pattern and Initial Values

LINEQS Model Statement

| | Matrix | Rows | Columns | -----Matrix Type----- | |
|--------|--------|---------|---------|-----------------------|-----------|
| Term 1 | 1 | _SEL_ | 8 | 18 | SELECTION |
| | 2 | _BETA_ | 18 | 18 | EQSBETA |
| | 3 | _GAMMA_ | 18 | 10 | EQSGAMMA |
| | 4 | _PHI_ | 10 | 10 | SYMMETRIC |

The 8 Endogenous Variables

| | X1 | X2 | X3 | X4 | X5 | X6 | X7 |
|----------|----|----|----|----|----|----|----|
| Manifest | | | | | | | |
| Latent | X8 | | | | | | |

The 10 Exogenous Variables

| | F1 | F2 | | | | | |
|----------|--------|--------|--------|--------|--------|--------|--------|
| Manifest | | | | | | | |
| Latent | delta1 | delta2 | delta3 | delta4 | delta5 | delta6 | delta7 |
| Error | delta8 | | | | | | |

Correlations

| | X1 | X2 | X3 | X4 | X5 | X6 | X7 | X8 |
|----|---------|---------|---------|--------|---------|---------|---------|---------|
| X1 | 1.0000 | 0.2564 | 0.2431 | 0.2183 | 0.0400 | -0.0195 | 0.0151 | -0.0521 |
| X2 | 0.2564 | 1.0000 | 0.3190 | 0.2945 | -0.0214 | -0.0184 | -0.0417 | -0.1343 |
| X3 | 0.2431 | 0.3190 | 1.0000 | 0.2107 | -0.0055 | -0.1010 | -0.0125 | -0.0164 |
| X4 | 0.2183 | 0.2945 | 0.2107 | 1.0000 | 0.0438 | 0.0173 | 0.1285 | 0.0530 |
| X5 | 0.0400 | -0.0214 | -0.0055 | 0.0438 | 1.0000 | 0.2790 | 0.2253 | 0.2549 |
| X6 | -0.0195 | -0.0184 | -0.1010 | 0.0173 | 0.2790 | 1.0000 | 0.1468 | 0.2926 |
| X7 | 0.0151 | -0.0417 | -0.0125 | 0.1285 | 0.2253 | 0.1468 | 1.0000 | 0.2662 |
| X8 | -0.0521 | -0.1343 | -0.0164 | 0.0530 | 0.2549 | 0.2926 | 0.2662 | 1.0000 |

Determinant 0.493401 Ln -0.706433

NOTE: Some initial estimates computed by approximative factor analysis.

Skipping listing of starting values ...

Levenberg-Marquardt Optimization

Scaling Update of More (1978)

Parameter Estimates 24
 Functions (Observations) 36
 Lower Bounds 8
 Upper Bounds 0

Optimization Start

Active Constraints 0 Objective Function 0.0920853206
 Max Abs Gradient Element 0.1649778516 Radius 1

| Iter | Rest arts | Func Calls | Act Con | Objective Function | Obj Fun Change | Max Abs Gradient Element | Lambda | Actual Over Pred Change |
|------|-----------|------------|---------|--------------------|----------------|--------------------------|---------|-------------------------|
| 1* | 0 | 2 | 0 | 0.04786 | 0.0442 | 0.0469 | 111E-16 | 0.866 |
| 2 | 0 | 3 | 0 | 0.04618 | 0.00168 | 0.00441 | 0 | 0.975 |
| 3* | 0 | 4 | 0 | 0.04614 | 0.000045 | 0.00173 | 111E-16 | 0.900 |
| 4* | 0 | 5 | 0 | 0.04613 | 3.403E-6 | 0.000272 | 111E-16 | 0.899 |
| 5 | 0 | 6 | 0 | 0.04613 | 2.489E-7 | 0.000173 | 0 | 0.826 |
| 6* | 0 | 7 | 0 | 0.04613 | 4.553E-8 | 0.000022 | 111E-16 | 0.949 |
| 7* | 0 | 8 | 0 | 0.04613 | 1.947E-9 | 9.409E-6 | 111E-16 | 0.911 |

Optimization Results

Iterations 7 Function Calls 9
 Jacobian Calls 8 Active Constraints 0
 Objective Function 0.0461331842 Max Abs Gradient Element 9.4087288E-6
 Lambda 1.110223E-14 Actual Over Pred Change 0.9105070821
 Radius 0.2773814698

ABSGCONV convergence criterion satisfied.

NOTE: Moore-Penrose inverse is used in covariance matrix.

NOTE: Covariance matrix for the estimates is not full rank.

NOTE: The variance of some parameter estimates is zero or some parameter estimates are linearly related to other parameter estimates as shown in the following equations:

$$\begin{aligned}
 \text{lambda21} &= 1.101807\text{E-}16 - 0.793600 * \text{lambda11} - 0.440518 * \text{lambda12} - 0.785901 \\
 & * \text{lambda22} - 0.839060 * \text{lambda31} \\
 & - 0.580388 * \text{lambda32} - 0.981720 * \text{lambda41} - 0.241288 \\
 & * \text{lambda42} - 0.545552 * \text{lambda51} \\
 & + 0.841570 * \text{lambda52} - 0.394371 * \text{lambda61} + 0.888376 \\
 & * \text{lambda62} - 0.501799 * \text{lambda71} \\
 & + 0.728809 * \text{lambda72} - 0.442683 * \text{lambda81} + 1.113665 \\
 & * \text{lambda82}
 \end{aligned}$$

Predicted Model Matrix

| | X1 | X2 | X3 | X4 | X5 | X6 | X7 | X8 |
|----|-------------|---------|---------|----------|---------|-----------|---------|---------|
| X1 | 1.0000 | 0.2755 | 0.2227 | 0.2140 | 0.0150 | -0.0189 | 0.0187 | -0.0337 |
| X2 | 0.2755 | 1.0000 | 0.3130 | 0.2831 | -0.0280 | -0.0734 | -0.0172 | -0.1045 |
| X3 | 0.2227 | 0.3130 | 1.0000 | 0.2329 | -0.0074 | -0.0446 | -0.0005 | -0.0664 |
| X4 | 0.2140 | 0.2831 | 0.2329 | 1.0000 | 0.0804 | 0.0418 | 0.0766 | 0.0401 |
| X5 | 0.0150 | -0.0280 | -0.0074 | 0.0804 | 1.0000 | 0.2327 | 0.2144 | 0.2849 |
| X6 | -0.0189 | -0.0734 | -0.0446 | 0.0418 | 0.2327 | 1.0000 | 0.2043 | 0.2813 |
| X7 | 0.0187 | -0.0172 | -0.0005 | 0.0766 | 0.2144 | 0.2043 | 1.0000 | 0.2499 |
| X8 | -0.0337 | -0.1045 | -0.0664 | 0.0401 | 0.2849 | 0.2813 | 0.2499 | 1.0000 |
| | Determinant | | | 0.516696 | Ln | -0.660300 | | |

Confirmatory and Exploratory factor analysis with simulated data 6
 Non-identified 2-factor model
 Try to fit non-identified 2-factor model with proc calis
 22:17 Wednesday, April 5, 2006

The CALIS Procedure
 Covariance Structure Analysis: Maximum Likelihood Estimation

| | |
|--|----------------|
| Fit Function | 0.0461 |
| Goodness of Fit Index (GFI) | 0.9888 |
| GFI Adjusted for Degrees of Freedom (AGFI) | 0.9665 |
| Root Mean Square Residual (RMR) | 0.0261 |
| Parsimonious GFI (Mulaik, 1989) | 0.4238 |
| Chi-Square | 23.0205 |
| Chi-Square DF | 12 |
| Pr > Chi-Square | 0.0276 |
| Independence Model Chi-Square | 352.51 |
| Independence Model Chi-Square DF | 28 |
| RMSEA Estimate | 0.0429 |
| RMSEA 90% Lower Confidence Limit | 0.0139 |
| RMSEA 90% Upper Confidence Limit | 0.0692 |
| ECVI Estimate | 0.1441 |
| ECVI 90% Lower Confidence Limit | 0.1243 |
| ECVI 90% Upper Confidence Limit | 0.1797 |
| Probability of Close Fit | 0.6363 |
| Bentler's Comparative Fit Index | 0.9660 |
| Normal Theory Reweighted LS Chi-Square | 22.5125 |
| Akaike's Information Criterion | -0.9795 |
| Bozdogan's (1987) CAIC | -63.5548 |
| Schwarz's Bayesian Criterion | -51.5548 |
| McDonald's (1989) Centrality | 0.9890 |
| Bentler & Bonett's (1980) Non-normed Index | 0.9208 |
| Bentler & Bonett's (1980) NFI | 0.9347 |
| James, Mulaik, & Brett (1982) Parsimonious NFI | 0.4006 |
| Z-Test of Wilson & Hilferty (1931) | 1.9184 |
| Bollen (1986) Normed Index Rho1 | 0.8476 |
| Bollen (1988) Non-normed Index Delta2 | 0.9676 |
| Hoelster's (1983) Critical N | 457 |

The CALIS Procedure
 Covariance Structure Analysis: Maximum Likelihood Estimation

Manifest Variable Equations with Estimates

| | | | | | | | | | |
|---------|---|----------|----------|---|---------|----------|---|--------|--------|
| X1 | = | -0.2166* | F1 | + | 0.3901* | F2 | + | 1.0000 | delta1 |
| Std Err | | 0.0512 | lambda11 | | 0.0563 | lambda12 | | | |
| t Value | | -4.2301 | | | 6.9247 | | | | |
| X2 | = | -0.3864* | F1 | + | 0.4916* | F2 | + | 1.0000 | delta2 |
| Std Err | | 0.0495 | lambda21 | | 0.0556 | lambda22 | | | |
| t Value | | -7.7996 | | | 8.8467 | | | | |
| X3 | = | -0.2853* | F1 | + | 0.4125* | F2 | + | 1.0000 | delta3 |
| Std Err | | 0.0507 | lambda31 | | 0.0555 | lambda32 | | | |
| t Value | | -5.6290 | | | 7.4326 | | | | |
| X4 | = | -0.1186* | F1 | + | 0.4826* | F2 | + | 1.0000 | delta4 |
| Std Err | | 0.0483 | lambda41 | | 0.0581 | lambda42 | | | |
| t Value | | -2.4568 | | | 8.3003 | | | | |
| X5 | = | 0.4137* | F1 | + | 0.2682* | F2 | + | 1.0000 | delta5 |
| Std Err | | 0.0556 | lambda51 | | 0.0522 | lambda52 | | | |
| t Value | | 7.4476 | | | 5.1393 | | | | |
| X6 | = | 0.4367* | F1 | + | 0.1939* | F2 | + | 1.0000 | delta6 |
| Std Err | | 0.0565 | lambda61 | | 0.0511 | lambda62 | | | |
| t Value | | 7.7329 | | | 3.7937 | | | | |
| X7 | = | 0.3583* | F1 | + | 0.2467* | F2 | + | 1.0000 | delta7 |
| Std Err | | 0.0554 | lambda71 | | 0.0534 | lambda72 | | | |
| t Value | | 6.4728 | | | 4.6164 | | | | |
| X8 | = | 0.5475* | F1 | + | 0.2176* | F2 | + | 1.0000 | delta8 |
| Std Err | | 0.0584 | lambda81 | | 0.0481 | lambda82 | | | |
| t Value | | 9.3775 | | | 4.5273 | | | | |

Variances of Exogenous Variables

| Variable | Parameter | Estimate | Standard Error | t Value |
|----------|-----------|----------|----------------|---------|
| F1 | | 1.00000 | | |
| F2 | | 1.00000 | | |
| delta1 | theta1 | 0.80088 | 0.06122 | 13.08 |
| delta2 | theta2 | 0.60903 | 0.06843 | 8.90 |
| delta3 | theta3 | 0.74843 | 0.06188 | 12.10 |
| delta4 | theta4 | 0.75299 | 0.06246 | 12.06 |
| delta5 | theta5 | 0.75689 | 0.06338 | 11.94 |
| delta6 | theta6 | 0.77167 | 0.06259 | 12.33 |
| delta7 | theta7 | 0.81077 | 0.06222 | 13.03 |
| delta8 | theta8 | 0.65288 | 0.06790 | 9.62 |

The FACTOR Procedure
 Initial Factor Method: Maximum Likelihood

Prior Communalities Estimates: SMC

| X1 | X2 | X3 | X4 |
|------------|------------|------------|------------|
| 0.11488172 | 0.19863374 | 0.15101564 | 0.14062067 |
| X5 | X6 | X7 | X8 |
| 0.13509327 | 0.14347560 | 0.11307973 | 0.17690305 |

Preliminary Eigenvalues: Total = 1.38529374 Average = 0.17316172

| | Eigenvalue | Difference | Proportion | Cumulative |
|---|------------|------------|------------|------------|
| 1 | 1.15802941 | 0.13336140 | 0.8359 | 0.8359 |
| 2 | 1.02466801 | 0.99242098 | 0.7397 | 1.5756 |
| 3 | 0.03224703 | 0.09301184 | 0.0233 | 1.5989 |
| 4 | -.06076481 | 0.02558495 | -0.0439 | 1.5550 |
| 5 | -.08634975 | 0.05936803 | -0.0623 | 1.4927 |
| 6 | -.14571778 | 0.07765969 | -0.1052 | 1.3875 |
| 7 | -.22337747 | 0.09006343 | -0.1612 | 1.2263 |
| 8 | -.31344090 | | -0.2263 | 1.0000 |

2 factors will be retained by the PROPORTION criterion.

| Iteration | Criterion | Ridge | Change | Communalities | | | |
|-----------|-----------|--------|--------|---------------|---------|---------|---------|
| 1 | 0.0462291 | 0.0000 | 0.1823 | 0.20296 | 0.38096 | 0.25435 | 0.24588 |
| | | | | 0.25187 | 0.22740 | 0.18489 | 0.34224 |
| 2 | 0.0461358 | 0.0000 | 0.0093 | 0.19973 | 0.39030 | 0.25188 | 0.24633 |
| | | | | 0.24371 | 0.22976 | 0.18923 | 0.34527 |
| 3 | 0.0461333 | 0.0000 | 0.0017 | 0.19917 | 0.39078 | 0.25168 | 0.24695 |
| | | | | 0.24343 | 0.22842 | 0.18904 | 0.34696 |
| 4 | 0.0461332 | 0.0000 | 0.0003 | 0.19913 | 0.39095 | 0.25158 | 0.24699 |
| | | | | 0.24313 | 0.22838 | 0.18923 | 0.34707 |

Convergence criterion satisfied.

Significance Tests Based on 500 Observations

| Test | DF | Chi-Square | Pr > ChiSq |
|--------------------------------|----|------------|------------|
| H0: No common factors | 28 | 350.0376 | <.0001 |
| HA: At least one common factor | | | |
| H0: 2 Factors are sufficient | 13 | 22.7975 | 0.0442 |
| HA: More factors are needed | | | |

| | |
|--|------------------|
| Chi-Square without Bartlett's Correction | 23.020460 |
| Akaike's Information Criterion | -2.979540 |
| Schwarz's Bayesian Criterion | -57.769445 |
| Tucker and Lewis's Reliability Coefficient | 0.934473 |

Squared Canonical Correlations

| | Factor1 | Factor2 |
|--|------------|------------|
| | 0.61304844 | 0.57492542 |

Eigenvalues of the Weighted Reduced Correlation Matrix: Total = 2.93683106 Average = 0.36710388

| | Eigenvalue | Difference | Proportion | Cumulative |
|---|------------|------------|------------|------------|
| 1 | 1.58430280 | 0.23177448 | 0.5395 | 0.5395 |
| 2 | 1.35252832 | 1.17146987 | 0.4605 | 1.0000 |
| 3 | 0.18105845 | 0.10067175 | 0.0617 | 1.0617 |
| 4 | 0.08038670 | 0.02220057 | 0.0274 | 1.0890 |
| 5 | 0.05818613 | 0.07874782 | 0.0198 | 1.1088 |
| 6 | -.02056169 | 0.09392355 | -0.0070 | 1.1018 |
| 7 | -.11448523 | 0.07009920 | -0.0390 | 1.0629 |
| 8 | -.18458444 | | -0.0629 | 1.0000 |

Factor Pattern

| | Factor1 | Factor2 |
|----|----------|---------|
| X1 | 0.38757 | 0.22115 |
| X2 | 0.58536 | 0.21979 |
| X3 | 0.45796 | 0.20458 |
| X4 | 0.35183 | 0.35102 |
| X5 | -0.21452 | 0.44397 |
| X6 | -0.27276 | 0.39237 |
| X7 | -0.17829 | 0.39679 |
| X8 | -0.35509 | 0.47010 |

Variance Explained by Each Factor

| Factor | Weighted | Unweighted |
|---------|------------|------------|
| Factor1 | 1.58430280 | 1.10466437 |
| Factor2 | 1.35252832 | 0.99177981 |

Final Communalities Estimates and Variable Weights
 Total Communality: Weighted = 2.936831 Unweighted = 2.096444

| Variable | Communality | Weight |
|----------|-------------|------------|
| X1 | 0.19912149 | 1.24864087 |
| X2 | 0.39095278 | 1.64189820 |
| X3 | 0.25157813 | 1.33614951 |
| X4 | 0.24699900 | 1.32799573 |
| X5 | 0.24312507 | 1.32123679 |
| X6 | 0.22835359 | 1.29597303 |
| X7 | 0.18923091 | 1.23338999 |
| X8 | 0.34708321 | 1.53154694 |

Confirmatory and Exploratory factor analysis with simulated data 10
 Non-identified 2-factor model
 Estimate Loadings with exploratory factor analysis
 22:17 Wednesday, April 5, 2006

The FACTOR Procedure
 Rotation Method: Varimax

Orthogonal Transformation Matrix

| | 1 | 2 |
|---|---------|----------|
| 1 | 0.86701 | -0.49829 |
| 2 | 0.49829 | 0.86701 |

Rotated Factor Pattern

| | Factor1 | Factor2 |
|----|----------|----------|
| X1 | 0.44623 | -0.00138 |
| X2 | 0.61703 | -0.10112 |
| X3 | 0.49899 | -0.05083 |
| X4 | 0.47995 | 0.12903 |
| X5 | 0.03523 | 0.49182 |
| X6 | -0.04097 | 0.47610 |
| X7 | 0.04314 | 0.43286 |
| X8 | -0.07363 | 0.58452 |

Variance Explained by Each Factor

| Factor | Weighted | Unweighted |
|---------|------------|------------|
| Factor1 | 1.52675593 | 1.06939360 |
| Factor2 | 1.41007520 | 1.02705058 |

Final Communalities Estimates and Variable Weights
 Total Communality: Weighted = 2.936831 Unweighted = 2.096444

| Variable | Communality | Weight |
|----------|-------------|------------|
| X1 | 0.19912149 | 1.24864087 |
| X2 | 0.39095278 | 1.64189820 |
| X3 | 0.25157813 | 1.33614951 |
| X4 | 0.24699900 | 1.32799573 |
| X5 | 0.24312507 | 1.32123679 |
| X6 | 0.22835359 | 1.29597303 |
| X7 | 0.18923091 | 1.23338999 |
| X8 | 0.34708321 | 1.53154694 |

Confirmatory and Exploratory factor analysis with simulated data 11
 Non-identified 2-factor model
 Start at rotated solution: Objective Function = 0.046133184?

Skipping a lot: The answer is *YES*.

Optimization Results

| | | | |
|--------------------|--------------|--------------------------|--------------|
| Iterations | 7 | Function Calls | 11 |
| Jacobian Calls | 8 | Active Constraints | 0 |
| Objective Function | 0.046133184 | Max Abs Gradient Element | 4.6322427E-6 |
| Lambda | 1.110223E-14 | Actual Over Pred Change | 0.9919197308 |
| Radius | 0.036444217 | | |

Skip some more: Loadings moved a bit from starting values, but not much.

Manifest Variable Equations with Estimates

| | | | | | | |
|---------|---|-----------------|---|-----------------|---|---------------|
| X1 | = | 0.4453*F1 | + | 0.0280*F2 | + | 1.0000 delta1 |
| Std Err | | 0.0572 lambda11 | | 0.0503 lambda12 | | |
| t Value | | 7.7895 | | 0.5574 | | |
| X2 | = | 0.6224*F1 | + | -0.0603*F2 | + | 1.0000 delta2 |
| Std Err | | 0.0605 lambda21 | | 0.0434 lambda22 | | |
| t Value | | 10.2860 | | -1.3892 | | |
| X3 | = | 0.5013*F1 | + | -0.0178*F2 | + | 1.0000 delta3 |
| Std Err | | 0.0577 lambda31 | | 0.0482 lambda32 | | |
| t Value | | 8.6856 | | -0.3704 | | |
| X4 | = | 0.4704*F1 | + | 0.1604*F2 | + | 1.0000 delta4 |
| Std Err | | 0.0569 lambda41 | | 0.0497 lambda42 | | |
| t Value | | 8.2649 | | 3.2253 | | |
| X5 | = | 0.00275*F1 | + | 0.4931*F2 | + | 1.0000 delta5 |
| Std Err | | 0.0477 lambda51 | | 0.0594 lambda52 | | |
| t Value | | 0.0576 | | 8.2989 | | |
| X6 | = | -0.0722*F1 | + | 0.4723*F2 | + | 1.0000 delta6 |
| Std Err | | 0.0484 lambda61 | | 0.0588 lambda62 | | |
| t Value | | -1.4925 | | 8.0318 | | |
| X7 | = | 0.0145*F1 | + | 0.4348*F2 | + | 1.0000 delta7 |
| Std Err | | 0.0498 lambda71 | | 0.0587 lambda72 | | |
| t Value | | 0.2918 | | 7.4112 | | |
| X8 | = | -0.1120*F1 | + | 0.5784*F2 | + | 1.0000 delta8 |
| Std Err | | 0.0446 lambda81 | | 0.0611 lambda82 | | |
| t Value | | -2.5132 | | 9.4658 | | |

Just show the unrotated and rotated loadings.

| | Factor Pattern | |
|----|----------------|---------|
| | Factor1 | Factor2 |
| X1 | 0.91601 | 0.22481 |
| X2 | 0.92820 | 0.22462 |
| X3 | 0.92296 | 0.22046 |
| X4 | 0.90313 | 0.24962 |
| X5 | -0.22649 | 0.92123 |
| X6 | -0.25341 | 0.90484 |
| X7 | -0.23431 | 0.90476 |
| X8 | -0.26506 | 0.91071 |

| | Rotated Factor Pattern | |
|----|------------------------|----------|
| | Factor1 | Factor2 |
| X1 | 0.94311 | -0.01269 |
| X2 | 0.95486 | -0.01594 |
| X3 | 0.94874 | -0.01865 |
| X4 | 0.93687 | 0.01456 |
| X5 | 0.01237 | 0.94858 |
| X6 | -0.01780 | 0.93948 |
| X7 | 0.00066 | 0.93461 |
| X8 | -0.02761 | 0.94810 |

And recall, these are the true parameter values:

```
/* True factor loadings (All communalities = .5^2 + .8^2 = 0.89) */  
L11 = .5; L12 = -.8;  
L21 = .5; L22 = -.8;  
L31 = .5; L32 = -.8;  
L41 = .5; L42 = -.8;  
L51 = .8; L52 = .5;  
L61 = .8; L62 = .5;  
L71 = .8; L72 = .5;  
L81 = .8; L82 = .5;
```